13.6 Linear Prediction and Linear Predictive Coding

We begin with a very general formulation that will allow us to make connections to various special cases. Let $\{y'_{\alpha}\}$ be a set of measured values for some underlying set of true values of a quantity y, denoted $\{y_{\alpha}\}$, related to these true values by the addition of random noise,

$$y'_{\alpha} = y_{\alpha} + n_{\alpha} \tag{13.6.1}$$

(compare equation 13.3.2, with a somewhat different notation). Our use of a Greek subscript to index the members of the set is meant to indicate that the data points are not necessarily equally spaced along a line, or even ordered: they might be "random" points in three-dimensional space, for example. Now, suppose we want to construct the "best" estimate of the true value of some particular point y_{\star} as a linear combination of the known, noisy, values. Writing

$$y_{\star} = \sum_{\alpha} d_{\star \alpha} y'_{\alpha} + x_{\star} \tag{13.6.2}$$

we want to find coefficients $d_{\star\alpha}$ that minimize, in some way, the *discrepancy* x_{\star} . The coefficients $d_{\star\alpha}$ have a "star" subscript to indicate that they depend on the choice of point y_{\star} . Later, we might want to let y_{\star} be one of the existing y_{α} 's. In that case, our problem becomes one of optimal filtering or estimation, closely related to the discussion in §13.3. On the other hand, we might want y_{\star} to be a completely new point. In that case, our problem will be one of *linear prediction*.

A natural way to minimize the discrepancy x_{\star} is in the statistical mean square sense. If angle brackets denote statistical averages, then we seek $d_{\star\alpha}$'s that minimize

$$\langle x_{\star}^{2} \rangle = \left\langle \left[\sum_{\alpha} d_{\star\alpha} (y_{\alpha} + n_{\alpha}) - y_{\star} \right]^{2} \right\rangle$$

$$= \sum_{\alpha\beta} (\langle y_{\alpha} y_{\beta} \rangle + \langle n_{\alpha} n_{\beta} \rangle) d_{\star\alpha} d_{\star\beta} - 2 \sum_{\alpha} \langle y_{\star} y_{\alpha} \rangle d_{\star\alpha} + \langle y_{\star}^{2} \rangle$$

$$(13.6.3)$$

Here we have used the fact that noise is uncorrelated with signal, e.g., $\langle n_{\alpha}y_{\beta}\rangle = 0$. The quantities $\langle y_{\alpha}y_{\beta}\rangle$ and $\langle y_{\star}y_{\alpha}\rangle$ describe the autocorrelation structure of the underlying data. We have already seen an analogous expression, (13.2.2), for the case of equally spaced data points on a line; we will meet correlation several times again in its statistical sense in Chapters 14 and 15. The quantities $\langle n_{\alpha}n_{\beta}\rangle$ describe the autocorrelation properties of the noise. Often, for point-to-point uncorrelated noise, we have $\langle n_{\alpha}n_{\beta}\rangle = \langle n_{\alpha}^2 \rangle \delta_{\alpha\beta}$. It is convenient to think of the various correlation quantities as comprising matrices and vectors,

$$\phi_{\alpha\beta} \equiv \langle y_{\alpha}y_{\beta} \rangle \qquad \phi_{\star\alpha} \equiv \langle y_{\star}y_{\alpha} \rangle \qquad \eta_{\alpha\beta} \equiv \langle n_{\alpha}n_{\beta} \rangle \text{ or } \langle n_{\alpha}^{2} \rangle \delta_{\alpha\beta} \quad (13.6.4)$$

Setting the derivative of equation (13.6.3) with respect to the $d_{\star\alpha}$'s equal to zero, one readily obtains the set of linear equations,

$$\sum_{\beta} \left[\phi_{\alpha\beta} + \eta_{\alpha\beta} \right] d_{\star\beta} = \phi_{\star\alpha} \tag{13.6.5}$$

Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America)

Sample page from NUMERICAL Copyright (C) 1986-1992 by Can

1986-1992 by Cambridge University Press. Programs Copyright (C)

RECIPES

IN FORTRAN

77: THE ART OF

SCIENTIFIC COMPUTING (ISBN 0-521

1986-1992 by Numerical Recipes

Software. -43064-X)

internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-his one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs

557

If we write the solution as a matrix inverse, then the estimation equation (13.6.2) becomes, omitting the minimized discrepancy x_{\star} ,

$$y_{\star} \approx \sum_{\alpha\beta} \phi_{\star\alpha} \left[\phi_{\mu\nu} + \eta_{\mu\nu} \right]_{\alpha\beta}^{-1} y_{\beta}'$$
(13.6.6)

From equations (13.6.3) and (13.6.5) one can also calculate the expected mean square value of the discrepancy at its minimum, denoted $\langle x_{\star}^2 \rangle_0$,

$$\left\langle x_{\star}^{2}\right\rangle_{0} = \left\langle y_{\star}^{2}\right\rangle - \sum_{\beta} d_{\star\beta}\phi_{\star\beta} = \left\langle y_{\star}^{2}\right\rangle - \sum_{\alpha\beta}\phi_{\star\alpha}\left[\phi_{\mu\nu} + \eta_{\mu\nu}\right]_{\alpha\beta}^{-1}\phi_{\star\beta} \qquad (13.6.7)$$

A final general result tells how much the mean square discrepancy $\langle x_{\star}^2 \rangle$ is increased if we use the estimation equation (13.6.2) not with the best values $d_{\star\beta}$, but with some other values $\hat{d}_{\star\beta}$. The above equations then imply

$$\left\langle x_{\star}^{2}\right\rangle = \left\langle x_{\star}^{2}\right\rangle_{0} + \sum_{\alpha\beta} (\widehat{d}_{\star\alpha} - d_{\star\alpha}) \left[\phi_{\alpha\beta} + \eta_{\alpha\beta}\right] (\widehat{d}_{\star\beta} - d_{\star\beta})$$
(13.6.8)

Since the second term is a pure quadratic form, we see that the increase in the discrepancy is only second order in any error made in estimating the $d_{\star\beta}$'s.

Connection to Optimal Filtering

If we change "star" to a Greek index, say γ , then the above formulas describe optimal filtering, generalizing the discussion of §13.3. One sees, for example, that if the noise amplitudes n_{α} go to zero, so likewise do the noise autocorrelations $\eta_{\alpha\beta}$, and, canceling a matrix times its inverse, equation (13.6.6) simply becomes $y_{\gamma} = y'_{\gamma}$. Another special case occurs if the matrices $\phi_{\alpha\beta}$ and $\eta_{\alpha\beta}$ are diagonal. In that case, equation (13.6.6) becomes

$$y_{\gamma} = \frac{\phi_{\gamma\gamma}}{\phi_{\gamma\gamma} + \eta_{\gamma\gamma}} y_{\gamma}' \tag{13.6.9}$$

which is readily recognizable as equation (13.3.6) with $S^2 \rightarrow \phi_{\gamma\gamma}$, $N^2 \rightarrow \eta_{\gamma\gamma}$. What is going on is this: For the case of equally spaced data points, and in the Fourier domain, autocorrelations become simply squares of Fourier amplitudes (Wiener-Khinchin theorem, equation 12.0.12), and the optimal filter can be constructed algebraically, as equation (13.6.9), without inverting any matrix.

More generally, in the time domain, or any other domain, an optimal filter (one that minimizes the square of the discrepancy from the underlying true value in the presence of measurement noise) can be constructed by estimating the autocorrelation matrices $\phi_{\alpha\beta}$ and $\eta_{\alpha\beta}$, and applying equation (13.6.6) with $\star \to \gamma$. (Equation 13.6.8 is in fact the basis for the §13.3's statement that even crude optimal filtering can be quite effective.)

Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America). Sample page from NUMERICAL RECIPES IN FORTRAN 77: THE ART OF Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyrid SCIENTIFIC COMPUTING (ISBN 0-521 -43064-X)

Linear Prediction

Classical *linear prediction* specializes to the case where the data points y_{β} are equally spaced along a line, y_i , i = 1, 2, ..., N, and we want to use M consecutive values of y_i to predict an M + 1st. Stationarity is assumed. That is, the autocorrelation $\langle y_j y_k \rangle$ is assumed to depend only on the difference |j - k|, and not on j or k individually, so that the autocorrelation ϕ has only a single index,

$$\phi_j \equiv \langle y_i y_{i+j} \rangle \approx \frac{1}{N-j} \sum_{i=1}^{N-j} y_i y_{i+j}$$
(13.6.10)

Here, the approximate equality shows one way to use the actual data set values to estimate the autocorrelation components. (In fact, there is a better way to make these estimates; see below.) In the situation described, the estimation equation (13.6.2) is

$$y_n = \sum_{j=1}^M d_j y_{n-j} + x_n \tag{13.6.11}$$

(compare equation 13.5.1) and equation (13.6.5) becomes the set of M equations for the M unknown d_i 's, now called the *linear prediction (LP) coefficients*,

$$\sum_{j=1}^{M} \phi_{|j-k|} d_j = \phi_k \qquad (k = 1, \dots, M)$$
(13.6.12)

Notice that while noise is not explicitly included in the equations, it is properly accounted for, *if* it is point-to-point uncorrelated: ϕ_0 , as estimated by equation (13.6.10) using *measured* values y'_i , actually estimates the diagonal part of $\phi_{\alpha\alpha} + \eta_{\alpha\alpha}$, above. The mean square discrepancy $\langle x_n^2 \rangle$ is estimated by equation (13.6.7) as

$$\langle x_n^2 \rangle = \phi_0 - \phi_1 d_1 - \phi_2 d_2 - \dots - \phi_M d_M$$
 (13.6.13)

To use linear prediction, we first compute the d_j 's, using equations (13.6.10) and (13.6.12). We then calculate equation (13.6.13) or, more concretely, apply (13.6.11) to the known record to get an idea of how large are the discrepancies x_i . If the discrepancies are small, then we can continue applying (13.6.11) right on into the future, imagining the unknown "future" discrepancies x_i to be zero. In this application, (13.6.11) is a kind of extrapolation formula. In many situations, this extrapolation turns out to be vastly more powerful than any kind of simple polynomial extrapolation. (By the way, you should not confuse the terms "linear prediction" and "linear extrapolation"; the general functional form used by linear prediction is *much* more complex than a straight line, or even a low-order polynomial!)

However, to achieve its full usefulness, linear prediction must be constrained in one additional respect: One must take additional measures to guarantee its *stability*. Equation (13.6.11) is a special case of the general linear filter (13.5.1). The condition that (13.6.11) be stable as a linear predictor is precisely that given in equations (13.5.5) and (13.5.6), namely that the characteristic polynomial

2

$$z^{N} - \sum_{j=1}^{N} d_{j} z^{N-j} = 0$$
(13.6.14)

Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America). Sample page from NUMERICAL RECIPES IN FORTRAN 77: THE ART OF Copyright (C) 1986-1992 by Cambridge University Press. Programs Convrin SCIENTIFIC COMPUTING (ISBN 0-521-43064-X)

have all N of its roots inside the unit circle,

$$|z| \le 1 \tag{13.6.15}$$

There is no guarantee that the coefficients produced by equation (13.6.12) will have this property. If the data contain many oscillations without any particular trend towards increasing or decreasing amplitude, then the complex roots of (13.6.14) will generally all be rather close to the unit circle. The finite length of the data set will cause some of these roots to be inside the unit circle, others outside. In some applications, where the resulting instabilities are slowly growing and the linear prediction is not pushed too far, it is best to use the "unmassaged" LP coefficients that come directly out of (13.6.12). For example, one might be extrapolating to fill a short gap in a data set; then one might extrapolate both forwards across the gap and backwards from the data beyond the gap. If the two extrapolations agree tolerably well, then instability is not a problem.

When instability *is* a problem, you have to "massage" the LP coefficients. You do this by (i) solving (numerically) equation (13.6.14) for its N complex roots; (ii) moving the roots to where you think they ought to be inside or on the unit circle; (iii) reconstituting the now-modified LP coefficients. You may think that step (ii) sounds a little vague. It is. There is no "best" procedure. If you think that your signal is truly a sum of undamped sine and cosine waves (perhaps with incommensurate periods), then you will want simply to move each root z_i onto the unit circle,

$$z_i \rightarrow z_i / |z_i| \tag{13.6.16}$$

In other circumstances it may seem appropriate to reflect a bad root across the unit circle

$$z_i \rightarrow 1/z_i^* \tag{13.6.17}$$

This alternative has the property that it preserves the amplitude of the output of (13.6.11) when it is driven by a sinusoidal set of x_i 's. It assumes that (13.6.12) has correctly identified the spectral width of a resonance, but only slipped up on identifying its time sense so that signals that should be damped as time proceeds end up growing in amplitude. The choice between (13.6.16) and (13.6.17) sometimes might as well be based on voodoo. We prefer (13.6.17).

Also magical is the choice of M, the number of LP coefficients to use. You should choose M to be as small as works for you, that is, you should choose it by experimenting with your data. Try M = 5, 10, 20, 40. If you need larger M's than this, be aware that the procedure of "massaging" all those complex roots is quite sensitive to roundoff error. Use double precision.

Linear prediction is especially successful at extrapolating signals that are smooth and oscillatory, though not necessarily periodic. In such cases, linear prediction often extrapolates accurately through *many cycles* of the signal. By contrast, polynomial extrapolation in general becomes seriously inaccurate after at most a cycle or two. A prototypical example of a signal that can successfully be linearly predicted is the height of ocean tides, for which the fundamental 12-hour period is modulated in phase and amplitude over the course of the month and year, and for which local Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America). Sample page from NUMERICAL RECIPES IN FORTRAN 77: THE Copyright (C) 1986-1992 by Cambridge University Press. Program: ART OF SCIENTIFIC COMPUTING (ISBN 0-521 -43064-X)

hydrodynamic effects may make even one cycle of the curve look rather different in shape from a sine wave.

We already remarked that equation (13.6.10) is not necessarily the best way to estimate the covariances ϕ_k from the data set. In fact, results obtained from linear prediction are remarkably sensitive to exactly how the ϕ_k 's are estimated. One particularly good method is due to Burg [1], and involves a recursive procedure for increasing the order M by one unit at a time, at each stage re-estimating the coefficients d_j , $j = 1, \ldots, M$ so as to minimize the residual in equation (13.6.13). Although further discussion of the Burg method is beyond our scope here, the method is implemented in the following routine [1,2] for estimating the LP coefficients d_j of a data set.

```
SUBROUTINE memcof(data,n,m,xms,d)
INTEGER m,n,MMAX,NMAX
REAL xms.d(m).data(n)
PARAMETER (MMAX=60,NMAX=2000)
   Given a real vector of data(1:n), and given m, this routine returns m linear prediction
   coefficients as d(1:m), and returns the mean square discrepancy as xms.
INTEGER i, j, k
REAL denom, p, pneum, wk1(NMAX), wk2(NMAX), wkm(MMAX)
if (m.gt.MMAX.or.n.gt.NMAX) pause 'workspace too small in memcof'
p=0.
do 11 j=1,n
   p=p+data(j)**2
enddo 11
xms=p/n
wk1(1)=data(1)
wk2(n-1)=data(n)
do 12 j=2,n-1
    wk1(j)=data(j)
    wk2(j-1)=data(j)
enddo 12
do 17 k=1.m
    pneum=0.
    denom=0.
    do 13 j=1,n-k
        pneum=pneum+wk1(j)*wk2(j)
        denom=denom+wk1(j)**2+wk2(j)**2
    enddo 13
    d(k)=2.*pneum/denom
    xms=xms*(1.-d(k)**2)
    do 14 i=1,k-1
        d(i) = wkm(i) - d(k) * wkm(k-i)
    enddo 14
      The algorithm is recursive, building up the answer for larger and larger values of m until
      the desired value is reached. At this point in the algorithm, one could return the vector
      d and scalar xms for a set of LP coefficients with k (rather than m) terms.
    if(k.eq.m)return
    do 15 i=1,k
        wkm(i)=d(i)
    enddo 15
    do 16 j=1,n-k-1
        wk1(j)=wk1(j)-wkm(k)*wk2(j)
        wk2(j)=wk2(j+1)-wkm(k)*wk1(j+1)
    enddo 16
enddo 17
pause 'never get here in memcof'
END
```

Sample page from NUMERICAL RECIPES IN FORTRAN 77: THE ART OF SCIENTIFIC COMPUTING (ISBN 0-521-43064-X) Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America).

Here are procedures for rendering the LP coefficients stable (if you choose to do so), and for extrapolating a data set by linear prediction, using the original or massaged LP coefficients. The routine zroots (§9.5) is used to find all complex roots of a polynomial.

```
SUBROUTINE fixrts(d,m)
INTEGER m, MMAX
REAL d(m)
PARAMETER (MMAX=100)
                                Largest expected value of m.
USES zroots
    Given the LP coefficients d(1:m), this routine finds all roots of the characteristic polynomial
    (13.6.14), reflects any roots that are outside the unit circle back inside, and then returns
    a modified set of coefficients d(1:m).
INTEGER i,j
LOGICAL polish
COMPLEX a(MMAX), roots(MMAX)
a(m+1)=cmplx(1.,0.)
                                Set up complex coefficients for polynomial root finder.
do 11 j=m,1,-1
    a(j)=cmplx(-d(m+1-j),0.)
enddo 11
polish=.true.
call zroots(a,m,roots,polish)
                                           Find all the roots.
do 12 j=1,m
                                           Look for a...
    if(abs(roots(j)).gt.1.)then
                                           root outside the unit circle,
        roots(j)=1./conjg(roots(j))
                                           and reflect it back inside.
    endif
enddo 12
a(1) = -roots(1)
                                Now reconstruct the polynomial coefficients,
a(2)=cmplx(1.,0.)
do 14 j=2,m
                                by looping over the roots
    a(j+1)=cmplx(1.,0.)
                                and synthetically multiplying.
    do 13 i=j,2,-1
        a(i)=a(i-1)-roots(j)*a(i)
    enddo 13
    a(1)=-roots(j)*a(1)
enddo 14
do 15 j=1,m
                                The polynomial coefficients are guaranteed to be real,
    d(m+1-j)=-real(a(j))
                                so we need only return the real part as new LP coefficients.
enddo 15
return
END
SUBROUTINE predic(data,ndata,d,m,future,nfut)
INTEGER ndata, nfut, m, MMAX
REAL d(m), data(ndata), future(nfut)
PARAMETER (MMAX=100)
    Given data(1:ndata), and given the data's LP coefficients d(1:m), this routine applies
    equation (13.6.11) to predict the next nfut data points, which it returns in the array
    future(1:nfut). Note that the routine references only the last m values of data, as
    initial values for the prediction.
   Parameter: MMAX is the largest expected value of m.
INTEGER j,k
REAL discrp, sum, reg(MMAX)
do 11 j=1,m
    reg(j)=data(ndata+1-j)
enddo 11
do 14 j=1,nfut
                                This is where you would put in a known discrepancy if you
    discrp=0.
    sum=discrp
                                    were reconstructing a function by linear predictive coding
    do 12 k=1,m
                                    rather than extrapolating a function by linear prediction.
                                   See text.
```

Sample page from NUMERICAL RECIPES IN FORTRAN 77: THE ART OF SCIENTIFIC COMPUTING (ISBN 0-521-43064-X) Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America).

С

```
sum=sum+d(k)*reg(k)
                       [If you want to implement circular arrays, you can avoid this
```

shifting of coefficients!]

enddo 12 do 13 k=m,2,-1 reg(k)=reg(k-1) enddo 13 reg(1)=sum future(j)=sum enddo 14 return END

Removing the Bias in Linear Prediction

You might expect that the sum of the d_i 's in equation (13.6.11) (or, more generally, in equation 13.6.2) should be 1, so that (e.g.) adding a constant to all the data points y_i yields a prediction that is increased by the same constant. However, the d_i 's do not sum to 1 but, in general, to a value slightly less than one. This fact reveals a subtle point, that the estimator of classical linear prediction is not unbiased, even though it does minimize the mean square discrepancy. At any place where the measured autocorrelation does not imply a better estimate, the equations of linear prediction tend to predict a value that tends towards zero.

Sometimes, that is just what you want. If the process that generates the y_i 's in fact has zero mean, then zero is the best guess absent other information. At other times, however, this behavior is unwarranted. If you have data that show only small variations around a positive value, you don't want linear predictions that droop towards zero.

Often it is a workable approximation to subtract the mean off your data set, perform the linear prediction, and then add the mean back. This procedure contains the germ of the correct solution; but the simple arithmetic mean is not quite the correct constant to subtract. In fact, an unbiased estimator is obtained by subtracting from every data point an autocorrelation-weighted mean defined by [3,4]

$$\overline{y} \equiv \sum_{\beta} \left[\phi_{\mu\nu} + \eta_{\mu\nu} \right]_{\alpha\beta}^{-1} y_{\beta} \Big/ \sum_{\alpha\beta} \left[\phi_{\mu\nu} + \eta_{\mu\nu} \right]_{\alpha\beta}^{-1}$$
(13.6.18)

With this subtraction, the sum of the LP coefficients should be unity, up to roundoff and differences in how the ϕ_k 's are estimated.

Linear Predictive Coding (LPC)

A different, though related, method to which the formalism above can be applied is the "compression" of a sampled signal so that it can be stored more compactly. The original form should be *exactly* recoverable from the compressed version. Obviously, compression can be accomplished only if there is redundancy in the signal. Equation (13.6.11) describes one kind of redundancy: It says that the signal, except for a small discrepancy, is predictable from its previous values and from a small number of LP coefficients. Compression of a signal by the use of (13.6.11) is thus called *linear predictive coding*, or LPC.

The basic idea of LPC (in its simplest form) is to record as a compressed file (i) the number of LP coefficients M, (ii) their M values, e.g., as obtained by memcof,

Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America). Sample page from NUMERICAL Copyright (C) 1986-1992 by Cam

RECIPES

IN FORTRAN

77: THE ART OF

SCIENTIFIC COMPUTING (ISBN 0-521

-43064-X)

(iii) the first M data points, and then (iv) for each subsequent data point only its residual discrepancy x_i (equation 13.6.1). When you are creating the compressed file, you find the residual by applying (13.6.1) to the previous M points, subtracting the sum from the actual value of the current point. When you are reconstructing the original file, you add the residual back in, at the point indicated in the routine predic.

It may not be obvious why there is any compression at all in this scheme. After all, we are storing one value of residual per data point! Why not just store the original data point? The answer depends on the relative sizes of the numbers involved. The residual is obtained by subtracting two very nearly equal numbers (the data and the linear prediction). Therefore, the discrepancy typically has only a very small number of nonzero bits. These can be stored in a compressed file. How do you do it in a high-level language? Here is one way: Scale your data to have integer values, say between ± 1000000 and -1000000 (supposing that you need six significant figures). Modify equation (13.6.1) by enclosing the sum term in an "integer part of" operator. The discrepancy will now, by definition, be an integer. Experiment with different values of M, to find LP coefficients that make the range of the discrepancy as small as you can. If you can get to within a range of ± 127 (and in our experience this is not at all difficult) then you can write it to a file as a single byte. This is a compression factor of 4, compared to 4-byte integer or floating formats.

Notice that the LP coefficients are computed using the *quantized* data, and that the discrepancy is also quantized, i.e., quantization is done both outside and inside the LPC loop. If you are careful in following this prescription, then, apart from the initial quantization of the data, you will not introduce even a single bit of roundoff error into the compression-reconstruction process: While the evaluation of the sum in (13.6.11) may have roundoff errors, the residual that you store is the value which, when added back to the sum, gives *exactly* the original (quantized) data value. Notice also that you do not need to massage the LP coefficients for stability; by adding the residual back in to each point, you never depart from the original data, so instabilities cannot grow. There is therefore no need for fixrts, above.

Look at §20.4 to learn about *Huffman coding*, which will further compress the residuals by taking advantage of the fact that smaller values of discrepancy will occur more often than larger values. A very primitive version of Huffman coding would be this: If most of the discrepancies are in the range ± 127 , but an occasional one is outside, then reserve the value 127 to mean "out of range," and then record on the file (immediately following the 127) a full-word value of the out-of-range discrepancy. §20.4 explains how to do much better.

There are many variant procedures that all fall under the rubric of LPC.

- If the spectral character of the data is time-variable, then it is best not to use a single set of LP coefficients for the whole data set, but rather to partition the data into segments, computing and storing different LP coefficients for each segment.
- If the data are really well characterized by their LP coefficients, and you can tolerate some small amount of error, then don't bother storing all of the residuals. Just do linear prediction until you are outside of tolerances, then reinitialize (using *M* sequential stored residuals) and continue predicting.
- In some applications, most notably speech synthesis, one cares only about the spectral content of the reconstructed signal, not the relative phases. In this case, one need not store any starting values at all, but only the

Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America). Sample page from NUMERICAL RECIPES IN FORTRAN Copyright (C) 1986-1992 by Cambridge University Press. F 77: THE ART OF SCIENTIFIC COMPUTING (ISBN 0-521 -43064-X

LP coefficients for each segment of the data. The output is reconstructed by driving these coefficients with initial conditions consisting of all zeros except for one nonzero spike. A speech synthesizer chip may have of order 10 LP coefficients, which change perhaps 20 to 50 times per second.

• Some people believe that it is interesting to analyze a signal by LPC, even when the residuals x_i are *not* small. The x_i 's are then interpreted as the underlying "input signal" which, when filtered through the all-poles filter defined by the LP coefficients (see §13.7), produces the observed "output signal." LPC reveals simultaneously, it is said, the nature of the filter *and* the particular input that is driving it. We are skeptical of these applications; the literature, however, is full of extravagant claims.

CITED REFERENCES AND FURTHER READING:

Childers, D.G. (ed.) 1978, *Modern Spectrum Analysis* (New York: IEEE Press), especially the paper by J. Makhoul (reprinted from *Proceedings of the IEEE*, vol. 63, p. 561, 1975).

Burg, J.P. 1968, reprinted in Childers, 1978. [1]

Anderson, N. 1974, reprinted in Childers, 1978. [2]

Cressie, N. 1991, in *Spatial Statistics and Digital Image Analysis* (Washington: National Academy Press). [3]

Press, W.H., and Rybicki, G.B. 1992, Astrophysical Journal, vol. 398, pp. 169–176. [4]

13.7 Power Spectrum Estimation by the Maximum Entropy (All Poles) Method

The FFT is not the only way to estimate the power spectrum of a process, nor is it necessarily the best way for all purposes. To see how one might devise another method, let us enlarge our view for a moment, so that it includes not only real frequencies in the Nyquist interval $-f_c < f < f_c$, but also the entire complex frequency plane. From that vantage point, let us transform the complex *f*-plane to a new plane, called the *z*-transform plane or *z*-plane, by the relation

$$z \equiv e^{2\pi i f \Delta} \tag{13.7.1}$$

where Δ is, as usual, the sampling interval in the time domain. Notice that the Nyquist interval on the real axis of the *f*-plane maps one-to-one onto the unit circle in the complex *z*-plane.

If we now compare (13.7.1) to equations (13.4.4) and (13.4.6), we see that the FFT power spectrum estimate (13.4.5) for any real sampled function $c_k \equiv c(t_k)$ can be written, except for normalization convention, as

$$P(f) = \left| \sum_{k=-N/2}^{N/2-1} c_k z^k \right|^2$$
(13.7.2)

Of course, (13.7.2) is not the *true* power spectrum of the underlying function c(t), but only an estimate. We can see in two related ways why the estimate is not likely to be exact. First, in the time domain, the estimate is based on only a finite range of the function c(t) which may, for all we know, have continued from $t = -\infty$ to ∞ . Second, in the z-plane of equation (13.7.2), the finite Laurent series offers, in general, only an approximation to a general analytic function of z. In fact, a formal expression for representing "true" power spectra (up to normalization) is

$$P(f) = \left|\sum_{k=-\infty}^{\infty} c_k z^k\right|^2 \tag{13.7.3}$$

Sample page from NUMERICAL RECIPES IN FORTRAN 77: THE ART OF SCIENTIFIC COMPUTING (ISBN 0-521-43064-X) Copyright (C) 1986-1992 by Cambridge University Press. Programs Copyright (C) 1986-1992 by Numerical Recipes Software. Permission is granted for internet users to make one paper copy for their own personal use. Further reproduction, or any copying of machine-readable files (including this one) to any server computer, is strictly prohibited. To order Numerical Recipes books, diskettes, or CDROMs visit website http://www.nr.com or call 1-800-872-7423 (North America only), or send email to trade@cup.cam.ac.uk (outside North America).